

## E X A M P L E 1

# Listing Selected Observations

### Featured Tools:

- PRINT procedure
- SORT procedure
- FORMAT procedure, VALUE statement
- WHERE statement
- FORMAT and LABEL statements

Selected Listing of Local Residential Properties Price Range \$100,000 to \$250,000								1
ZONE	TYPE	ADDRESS	PRICE	BEDROOM	BATH	SQFEET	AGE	
Roebuck	twostory	Lot 10 Red Coach Rd.	\$225,000	4	2.5	3,446	0	
North Ridge	twostory	4000 Skipjack Ct.	\$214,900	3	2.5	2,750	0	
Westend	twostory	5521 Horseshoe Circle	\$199,200	3	2.5	2,863	7	
East Lake	split	7141 Eastridge	\$198,000	4	2.5	2,600	10	
East Lake	ranch	9354 Sauls Rd.	\$197,000	3	2.5	2,441	1	
East Lake	ranch	9356 Sauls Rd.	\$197,000	3	2.5	2,441	1	
East Lake	twostory	1641 Pricewood Lane	\$195,000	3	2.5	1,940	4	
Inside Beltline	twostory	114 Grey Horse Dr.	\$189,900	4	2.5	2,584	0	
Roebuck	twostory	2103 Carriage Way	\$189,900	4	2.5	2,608	0	
Inside Beltline	townhouse	765 Crabtree Crossing	\$184,000	3	2.0	1,471	1	
Inside Beltline	split	101 Meadowglade Ln.	\$179,950	3	2.0	2,004	0	
Inside Beltline	townhouse	108 Chattle Close	\$179,900	3	2.5	2,080	4	
North Ridge	twostory	1800 Bridgeport	\$169,900	4	2.5	2,700	7	
East Lake	twostory	211 Whitehall Way	\$154,900	4	2.5	2,538	6	
North Ridge	twostory	6008 Brass Lantern Ct.	\$144,500	3	2.5	2,416	6	
Inside Beltline	twostory	100 Cumberland Green	\$143,000	3	2.5	1,650	0	
East Lake	ranch	8122 Maude Steward Rd.	\$129,900	3	3.0	2,400	2	
Roebuck	split	110 Skylark Way	\$123,500	3	2.0	1,976	10	
Westend	twostory	1532 St. Mary's Rd.	\$118,500	3	2.5	1,900	4	
Westend	split	603 Greentree Dr.	\$117,800	3	2.0	1,533	5	
East Lake	ranch	6509 Orchard Knoll	\$107,900	3	2.0	1,526	6	
Southside	townhouse	154 Montrose	\$102,000	2	2.0	1,595	6	
Mountain Brook	duplex	108 South Elm St.	\$100,000	3	1.0	1,569	73	
North Ridge	split	6324 Lakeland, Lake Park	\$100,000	3	2.0	1,662	12	

## Overview

This report shows a partial listing of a data set that describes houses for sale. The listing contains information about houses within a certain price range.

## Method

PROC SORT sorts the data set by the descending order of the sale price. PROC FORMAT creates a format that assigns a meaningful name for the geographic region of each house. PROC PRINT selects and prints observations based on the price of the house.

## Key Steps

- Sort the data by the descending order of PRICE.
- Create a format that assigns a meaningful description for each value of ZONE.
- Select the observations to print.
- Print only selected variables.

---

## The Program

**Set options.**

```
options nodate number pageno=1 linesize=132 pagesize=60;
```

**Create the HOUSING data set.**

HOUSING contains descriptive data about homes currently for sale. The INPUT statement reads two records to create each observation. Each value for ZONE represents a geographic area. The complete DATA step is in the appendix.

```
data housing;
  input zone $1. +1 type : $8. bedr bath sqfeet age
        schools : $15. / address & $25. price;
  cards;
4 twostory 4 2.5 2538 6 920:340/400/368
211 Whitehall Way 154900
1 twostory 4 2.5 2700 7 920:470/360/552
1800 Bridgeport 169900
3 twnhouse 2 2 1595 6 320:332/366/312
154 Montrose 102000
.
. more data lines
.
1 twostory 3 2.5 2416 6 920:568/356/318
6008 Brass Lantern Ct. 144500
4 ranch 3 2.5 2441 1 920:540/512/436
9356 Sauls Rd. 197000
1 twostory 5 3.5 4850 3 920:540/512/436
9317 Sauls Rd. 339950
4 split 3 1.5 1225 18 920:328/312/316
6424 Old Jenks Rd. 81900
;
;
```

**Sort the data values by the descending values of PRICE.**

```
proc sort data=housing;
  by descending price;
run;
```

**Create a format for ZONE that provides meaningful descriptions for geographic regions.**

```
proc format;
  value $zonefmt '1'='North Ridge'
                '2'='Inside Beltline'
                '3'='Southside'
                '4'='East Lake'
                '5'='Westend'
                '6'='Mountain Brook'
                '7'='Ensley'
                '8'='Roebuck';
run;
```

**Print the sorted data set.** UNIFORM formats all pages of a multipage report uniformly. LABEL uses variable labels as column headers. NOOBS suppresses the observation numbers.

```
proc print data=housing uniform label noobs;
```

**Specify the observations and variables to print.** The WHERE statement prints only those observations where PRICE is between 100000 and 250000, inclusive. The VAR statement specifies the variables to print and the order in which to print them.

```
where price between 100000 and 250000;
var zone type address price bedr bath sqfeet age;
```

**Associate formats and labels with variables.**

```
format sqfeet comma5. price dollar9. zone $zonefmt.;
label bedr='BEDROOM';
```

**Specify titles.**

```
title 'Selected Listing of Local Residential Properties';
title2 'Price Range $100,000 to $250,000';
run;
```

---

## Where to Go from Here

- **PROC PRINT reference and usage information.** For complete reference documentation, see Chapter 27, “The PRINT Procedure,” in the *SAS Procedures Guide, Version 6, Third Edition* and Chapter 32, “The PRINT Procedure,” in SAS Technical Report P-222, *Changes and Enhancements to Base SAS Software, Release 6.07*. For more usage information, see Chapter 24, “Printing Detail Reports,” in *SAS Language and Procedures: Usage, Version 6, First Edition*.
- **WHERE statement processing.** For more information, see Chapter 9, “SAS Language Statements,” in *SAS Language: Reference, Version 6, First Edition* and Chapter 14, “Reporting on Subsets of SAS Data Sets,” in *SAS Language and Procedures: Usage 2, Version 6, First Edition*.
- **Reading multiple records to create one observation.** For more information, see Chapter 3, “Starting with Raw Data,” in *SAS Language and Procedures: Usage*.
- **Routing output.** For more information on routing SAS output, see the SAS companion for your operating system. You can use the following tools in SAS software: PROC PRINTTO or the display manager FILE command. You can also use operating system commands, options, and statements.